In[8]:=
```
(*Video 1: Continuity*)
TraditionalForm[HoldForm[f[x] = Piecewise[{{x^2 + 1, x < 0}, {x + c, x ≥ 0}}]]]
```

Out[8]//TraditionalForm=

$$f(x) = \begin{cases} x^2 + 1 & x < 0 \\ x + c & x \geq 0 \end{cases}$$

In[43]:=
```
Manipulate[Plot[Piecewise[{{x^2 + 1, x < 0}, {x + c, x > 0}}],
  {x, -3, 3}, PlotRange → {-4, 4}, Exclusions → x == 0], {c, -3, 3}]
```

Out[43]=



In[51]:=
```
Limit[x^2 + 1, x → 0, Direction → "FromBelow"]
Limit[x + c, x → 0, Direction → "FromAbove"]
```

Out[51]= 1

Out[52]= c

```
(*These lateral limits match if and only if c=1. Thus,
c=1 is the only value that makes f[x] continuous at x=0.*)
```

In[44]:=
```
TraditionalForm[
  HoldForm[f[x] = Piecewise[{{c x^2 + 2 c x + 1, x < 1}, {c x - 2 c, x ≥ 1}}]]]
Manipulate[Plot[Piecewise[{{c x^2 + 2 c x + 1, x < 1}, {c x - 2 c, x ≥ 1}}],
   {x, -5, 5}, PlotRange → {-6, 6}, Exclusions → x == 1], {c, -3, 3}]
```

Out[44]//TraditionalForm=

$$f(x) = \begin{cases} c\,x^2 + 2\,c\,x + 1 & x < 1 \\ c\,x - 2\,c & x \geq 1 \end{cases}$$

Out[45]=



In[53]:=
```
Limit[c x^2 + 2 c x + 1, x → 1, Direction → "FromBelow"]
Limit[c x - 2 c, x → 1, Direction → "FromAbove"]
```

Out[53]= $1 + 3\,c$

Out[54]= $-c$

In[55]:=
```
Solve[1 + 3 c == -c, c]
```

Out[55]= $\left\{\left\{c \to -\frac{1}{4}\right\}\right\}$

```
(*The unique value of c such that f[x] is continuous at x=1 is c=-1/4.*)

(*Video 2: Infinite limits*)
```

In[57]:=
```
TraditionalForm[HoldForm[Limit[(x^3 + 2 x + 1) / (2 x^3 - 5 x + 3), x → Infinity]]]
```

Out[57]//TraditionalForm=

$$\lim_{x\to\infty} \frac{x^3 + 2x + 1}{2x^3 - 5x + 3}$$

In[88]:=
```
(*Numerator*)
x^3 (1 + 2 / x^2 + 1 / x^3)
Simplify[x^3 (1 + 2 / x^2 + 1 / x^3)]
```

Out[88]= $\left(1 + \frac{1}{x^3} + \frac{2}{x^2}\right) x^3$

Out[89]= $1 + 2x + x^3$

In[90]:=
```
(*Denominator*)
x^3 (2 - 5 / x^2 + 3 / x^3)
Simplify[x^3 (2 - 5 / x^2 + 3 / x^3)]
```

Out[90]= $\left(2 + \frac{3}{x^3} - \frac{5}{x^2}\right) x^3$

Out[91]= $3 - 5x + 2x^3$

In[93]:=
```
TraditionalForm[HoldForm[x^3 (1 + 2 / x^2 + 1 / x^3) / (x^3 (2 - 5 / x^2 + 3 / x^3))]]
TraditionalForm[HoldForm[(1 + 2 / x^2 + 1 / x^3) / ((2 - 5 / x^2 + 3 / x^3))]]
```

Out[93]//TraditionalForm=

$$\frac{x^3\left(1 + \frac{2}{x^2} + \frac{1}{x^3}\right)}{x^3\left(2 - \frac{5}{x^2} + \frac{3}{x^3}\right)}$$

Out[94]//TraditionalForm=

$$\frac{1 + \frac{2}{x^2} + \frac{1}{x^3}}{2 - \frac{5}{x^2} + \frac{3}{x^3}}$$

In[95]:=
```
Limit[(x^3 + 2 x + 1) / (2 x^3 - 5 x + 3), x → Infinity]
```

Out[95]= $\frac{1}{2}$

In[101]:= `Plot[{(x^3 + 2 x + 1) / (2 x^3 - 5 x + 3), 1 / 2}, {x, 1, 100}, PlotRange → {0, 1}]`

Out[101]=



`(*y=1/2 is a horizontal asymptote for this function as x-->Infinity.*)`

In[62]:= `TraditionalForm[HoldForm[Limit[(x^3 + 2 x + 1) / (x^5 - 2 x + 1), x → Infinity]]]`

Out[62]//TraditionalForm=

$$\lim_{x \to \infty} \frac{x^3 + 2x + 1}{x^5 - 2x + 1}$$

In[102]:= `Limit[(x^3 + 2 x + 1) / (x^5 - 2 x + 1), x → Infinity]`

Out[102]= 0

In[109]:= `Plot[{(x^3 + 2 x + 1) / (x^5 - 2 x + 1), 0}, {x, 1, 30}, PlotRange → {-1, 1}]`

Out[109]=



`(*y=0 is a horizontal asymptote for this function as x→ Infinity.*)`

In[64]:= `TraditionalForm[HoldForm[Limit[(x^6 + x + 10) / (x^4 - 2 x + 1), x → Infinity]]]`

Out[64]//TraditionalForm=

$$\lim_{x \to \infty} \frac{x^6 + x + 10}{x^4 - 2x + 1}$$

In[110]:= `Limit[(x^6 + x + 10) / (x^4 - 2 x + 1), x → Infinity]`

Out[110]= ∞

In[113]:=
```
Plot[(x^6 + x + 10) / (x^4 - 2 x + 1), {x, 1, 1000}]
```

Out[113]=




```
(*Video 3: Asymptotes*)
```

In[149]:=
```
graph = Plot[1 / (x - 3), {x, -4, 10}, PlotRange → 10]
```

Out[149]=



In[163]:=
```
Limit[1 / (x - 3), x → 3, Direction → "FromBelow"]
Limit[1 / (x - 3), x → 3, Direction → "FromAbove"]
```

Out[163]= $-\infty$

Out[164]= ∞

In[166]:=
```
Vasymp = ParametricPlot[{3, y}, {y, -10, 10}, PlotStyle → Red];
(* x=3 is a vertical asymptote for this function*)
Hasymp = ParametricPlot[{x, 0}, {x, -10, 10}, PlotStyle → Green];
(* y=0 is a horizontal asymptote for this function*)
Show[graph, Vasymp, Hasymp]
```

Out[168]=



In[170]:=
```
Manipulate[Plot[1 / (x - a) + b, {x, -4, 10}, PlotRange → 10, Exclusions → x == a],
 {a, 0, 5}, {b, -5, 5}]
```

Out[170]=



In[175]:=
```
Plot[{1 / 2 + 1 / (x^2 + 1), 1 / 2}, {x, -60, 60}, PlotRange → 3]
```

Out[175]=



In[172]:=
```
Limit[1 / 2 + 1 / (x^2 + 1), x → Infinity]
Limit[1 / 2 + 1 / (x^2 + 1), x → -Infinity]
```

Out[172]= $\frac{1}{2}$

Out[173]= $\frac{1}{2}$

```
(*y=1/2 is a horizontal asymptote for this function.*)
```

In[194]:=
```
TraditionalForm[HoldForm[x / 3 + 5 / (x^2 + 3) + 10]]
```

Out[194]//TraditionalForm=

$$\frac{x}{3} + \frac{5}{x^2 + 3} + 10$$

In[197]:=
```
Limit[x / 3 + 5 / (x^2 + 3) + 10, x → Infinity]
```

Out[197]= ∞

In[193]:=
```
Plot[{x / 3 + 5 / (x^2 + 3) + 10, x / 3 + 10}, {x, -50, 50}, PlotRange → All]
```

Out[193]=



In[198]:=
```
Limit[(x / 3 + 5 / (x^2 + 3) + 10) - (x / 3 + 10), x → Infinity]
Limit[(x / 3 + 5 / (x^2 + 3) + 10) - (x / 3 + 10), x → -Infinity]
```

Out[198]= 0

Out[199]= 0

```
(* y = x/3+10 is a (slanted) asymptote for this function: not horizontal,
nor vertical.*)
```